

# Competing Yeast Species

Kurt Bryan and SIMIODE

A worksheet to facilitate analyzing the competing yeast species project in Section 7.5.3 of the text. The goal is to use the data of Table 7.2 to estimate the parameters  $r_1$ ,  $K_1$ ,  $r_2$ ,  $K_2$ ,  $a$ , and  $b$  in equations (7.3)-(7.4).

```
> restart;  
with( Optimization ) :  
with( plots ) :
```

## The Data:

Time (hours) for yeast populations, run 1 (first 11 lines of Table 7.2).

```
> times1 := [6, 16, 24, 29, 40, 48, 53, 72, 93, 117, 141] :
```

And for run 2 (remaining 7 lines in Table 7.2)

```
> times2 := [7.5, 15, 24, 31.5, 33, 44, 51.5] :
```

Saccharomyces population, run 1 (column 2 in Table 7.2, with "nd" for "no data").

```
> sacc1 := [0.37, 8.87, 10.66, 12.5, 13.27, 12.87, 12.70, nd, nd, nd, nd] :
```

Saccharomyces population, run 2, (column 2, bottom 7 lines in Table 7.2)

```
> sacc2 := [1.63, 6.2, 10.97, 12.60, 12.9, 12.77, 12.9] :
```

Schizosaccharomyces population run 1, (column 4 in Table 7.2, with "nd" for "no data").

```
> schiz1 := [nd, 1, nd, 1.7, nd, 2.73, nd, 4.87, 5.67, 5.8, 5.83] :
```

Schizosaccharomyces population run 2, (column 4, bottom 7 lines in Table 7.2)

```
> schiz2 := [nd, 1.27, nd, 2.33, nd, nd, 4.56] :
```

Saccharomyces population, run 1, mixed Sach/Schiz experiment (column 3, first 11 lines)

```
> saccmix1 := [0.375, 3.99, 4.69, 6.15, nd, 7.27, 8.3, nd, nd, nd, nd] :
```

Saccharomyces population, run 2, mixed Sach/Schiz experiment (column 3, bottom 7 lines)

```
> saccmix2 := [0.923, 3.082, 5.78, 9.91, 9.47, 10.57, 9.883] :
```

Schizosaccharomyces population run 1, mixed Sach/Schiz experiment (column 5, first 11 lines)

```
> schizmix1 := [0.291, 0.98, 1.47, 1.46, nd, 1.71, 1.84, nd, nd, nd, nd] :
```

Schizosaccharomyces population run 2, mixed Sach/Schiz experiment (column 5, bottom 7 lines)

```
> schizmix2 := [0.371, 0.63, 1.22, 1.112, 1.225, 1.102, 0.961] :
```

Let's form amalgamated data sets for runs 1 and 2, for the Saccharomyces population alone (column 2), the Schizosaccharomyces population alone (column 4), the Saccharomyces population in competition (column 3), and the Schizosaccharomyces population in competition (column 5).

The data will be in (time, population) pairs.

```
> N1 := nops(times1) : N2 := nops(times2) :
```

```
> saccind := [seq( [times1[k], sacc1[k]], k = 1 ..N1), seq( [times2[k], sacc2[k]], k = 1 ..N2) ] :
```

```
> schizind := [seq( [times1[k], schiz1[k]], k = 1 ..N1), seq( [times2[k], schiz2[k]], k = 1 ..N2) ] :
```

```
> saccmix := [seq( [times[k], saccmix1[k]], k = 1 ..N1), seq( [times2[k], saccmix2[k]], k = 1 ..N2) ] :
```

```
> schizmix := [seq( [times1[k], schizmix1[k]], k = 1 ..N1), seq( [times2[k], schizmix2[k]], k = 1 ..N2) ] :
```

We can plot the data. For example, for the saccharomyces population growing in isolation, with all "nd" points replaced by -1 and the view restricted to nonnegative population values so those points

don't show up.

```
> pointplot(subs(nd=-1, saccind), color=blue, symbol=solidcircle, symbolsize=15, view=[0
    ..60, 0..15])
```

A similar plot can be constructed for the schizosaccharomyces population in isolation, or either species in competition.

**Estimate Growth Parameters for Saccharomyces:** We can estimate the parameters  $r_1$  and  $K_1$  for the logistic growth model for the saccharomyces population using the corresponding data.

The logistic ODE is

```
> de := u'(t) = r1·u(t) · (1 - u(t)/K1) :
```

The initial data could be taken as any (time, population) data point, but we'll use the first:

```
> t0 := saccind[1, 1]; u10 := saccind[1, 2]; #initial time and population
```

Solve the logistic ODE with this data

```
> u1sol := rhs(dsolve({de, u(t0)=u10}, u(t)))
```

A plot or even a glance at the data suggests that  $K_1$  is around 12 to 14. To estimate  $K_1$  and  $r_1$  more carefully, form a sum of squares that ignores missing data

```
> SS := 0 : #Will hold sum of squares
N0 := N1 + N2 : #Total number of data points
for k from 1 to N0 do
    if type(saccind[k][2], numeric) then #Check if data is defined or "nd"
        SS := SS + (subs(t=saccind[k][1], u1sol) - saccind[k][2])^2 :
    fi:
od:
```

Minimize the sum of squares to obtain a best fit for  $r_1$  and  $K_1$ , using Maple's "Minimize" command. Initial guess is  $K_1 = 12$ ,  $r_1 = 0.5$ .

```
> sol1 := Minimize(SS, initialpoint = {K1 = 12, r1 = 0.5})
```

Define the best-fit solution:

```
> ubest1 := subs(sol1[2], u1sol)
```

Compare the plot and the data on an appropriate range (up to  $t = 60$  here).

```
> plot1 := plot(ubest1, t=0..60, color=red) :
> plot2 := pointplot(subs(nd=-1, saccind), color=blue, symbol=solidcircle, symbolsize=15,
    view=[0..60, 0..15]) :
display(plot1, plot2);
```

A similar procedure can be used to estimate  $r_2$  and  $K_2$  for the schizosaccharomyces data, using (for example) the data point  $t_0 = 16$ ,  $u_0 = 1$  as an "initial" data point.

**The Competition Parameters a and b:** The ODEs that govern the competing species system are

```
> de1 := u1'(t) = r1·u1(t) · (1 - u1(t)/K1 - a/K1 · u2(t));
    de2 := u2'(t) = r2·u2(t) · (1 - u2(t)/K2 - b/K2 · u1(t))
```

A closed-form solution cannot be obtained, but a guess-and-plot approach based on solving these ODEs numerically can work, using the values for  $r_1$ ,  $K_1$ ,  $r_2$ ,  $K_2$  found above.

```
>
```